

WEATHER BUREAU
Office of Systems Development
Techniques Development Laboratory
Silver Spring, Md.

July 1969

An Operationally Oriented Objective Analysis Program



Technical Memorandum WBTM TDL 22

U.S. DEPARTMENT OF COMMERCE / ENVIRONMENTAL SCIENCE SERVICES ADMINISTRATION

ESSA TECHNICAL MEMORANDA

Weather Bureau, Techniques Development Laboratory Series

The primary purpose of the Techniques Development Laboratory of the Office of Systems Development is to translate increases in basic knowledge in meteorology and allied disciplines into improved operating techniques and procedures. To achieve this goal, TDL conducts and sponsors applied research and development aimed at improvement of diagnostic and prognostic methods for producing weather information. The laboratory carries out studies both for the general improvement of prediction methodology used in the National Meteorological Service System and for more effective utilization of weather forecasts by the ultimate user.

ESSA Technical Memoranda in the Weather Bureau Techniques Development Laboratory series facilitate rapid distribution of material which may be preliminary in nature and which may be published formally elsewhere at a later date. The first five papers in the TDL series are part of the former Weather Bureau Technical Notes series.

Papers listed below are available from the Clearinghouse for Federal Scientific and Technical Information, U.S. Department of Commerce, Sills Bldg., 5285 Port Royal Road, Springfield, Va. 22151. Price: \$3.00 hard copy; \$0.65 microfiche. Order by accession number shown in parentheses at end of each entry.

- TN 10 TDL 1 Objective Prediction of Daily Surface Temperature. William H. Klein, Curtis W. Crockett, and Carlos R. Dunn, October 1965. (PB-168 590)
- TN 11 TDL 2 Hurricane Cindy Galveston Bay Tides. N. A. Pore, A. T. Angelo, and J. G. Taylor, September 1965. (PB-168 608)
- TN 29 TDL 3 Atmospheric Effects on Re-Entry Vehicle Dispersions. Karl R. Johannessen, December 1965. (PB-169 381)
- TN 45 TDL 4 A Synoptic Climatology of Winter Precipitation from 700-mb. Lows for the Intermountain Areas of the West. D. L. Jorgensen, W. H. Klein, and A. F. Korte, May 1966. (PB-170 635)
- TN 47 TDL 5 Hemispheric Specification of Sea Level Pressure from Numerical 700-mb. Height Forecasts. William H. Klein and Billy M. Lewis, June 1966. (PB-173 091)
- WBTM TDL 6 A Fortran Program for the Calculation of Hourly Values of Astronomical Tide and Time and Height of High and Low Water. N. A. Pore and R. A. Cummings, January 1967. (PB-174 660)
- WBTM TDL 7 Numerical Experiments Leading to the Design of Optimum Global Meteorological Networks. M. A. Alaka and F. Lewis, February 1967. (PB-174 497)
- WBTM TDL 8 An Experiment in the Use of the Balance Equation in the Tropics. M. A. Alaka, D. T. Rubsam, and G. E. Fisher, March 1967. (PB-174 501)
- WBTM TDL 9 A Survey of Studies of Aerological Network Requirements. M. A. Alaka, May 1967. (PB-174 984)
- WBTM TDL 10 Objective Determination of Sea Level Pressure from Upper Level Heights. William Klein, Frank Lewis, and John Stackpole, May 1967. (PB-179 949)
- WBTM TDL 11 Short Range, Subsynoptic Surface Weather Prediction. H. R. Glahn and D. A. Lowry, July 1967. (PB-175 772)
- WBTM TDL 12 Charts Giving Station Precipitation in the Plateau States from 700-Mb. Lows During Winter. D. L. Jorgensen, A. F. Korte, and J. A. Bunce, Jr., October 1967. (PB-176 742)
- WBTM TDL 13 Interim Report on Sea and Swell Forecasting. N. A. Pore and W. S. Richardson, December 1967. (PB-177 038)
- WBTM TDL 14 Meteorological Analysis of 1964-65 ICAO Turbulence Data. DeVer Colson, September 1968. (PB-180 268)
- WBTM TDL 15 Prediction of Temperature and Dew Point by Three-Dimensional Trajectories. Ronald M. Reap, September 1968. (PB-180 727)

(Continued on inside back cover)

U.S. DEPARTMENT OF COMMERCE
Environmental Science Services Administration
Weather Bureau

ESSA Technical Memorandum WBTM TDL 22

AN OPERATIONALLY ORIENTED OBJECTIVE
ANALYSIS PROGRAM

Harry R. Glahn, George W. Hollenbaugh, and Dale A. Lowry



OFFICE OF SYSTEMS DEVELOPMENT
TECHNIQUES DEVELOPMENT LABORATORY

SILVER SPRING, MD.
July 1969

UDC 551.509.25:681.3.06
551.5 Meteorology
.509 Synoptic analysis
.25 Prognostic charts
681.3 Automatic data
processing
.06 Computer programs

CONTENTS

	Page
Abstract	1
Introduction	1
Analysis Area and Data	1
The Analysis Method	3
Error-Detecting Procedure	4
Smoothing	5
Sea-Level Pressure Analysis	5
Saturation Deficit Analysis	8
References	.12
Appendix	.13

AN OPERATIONALLY ORIENTED OBJECTIVE ANALYSIS PROGRAM

Harry R. Glahn, George W. Hollenbaugh, and Dale A. Lowry

ABSTRACT

A computer program which performs objective analyses of sea-level pressure and saturation deficit by the method of successive approximations is described. The program was designed to be run operationally and includes error detecting routines. Analyses of hourly airways data collected on magnetic tape are made on a 39 x 40 grid over the eastern and central U.S. The small-scale detail specified by the relatively dense observations is retained with a grid-length of about 50 miles. Details of the error-detecting, correction, and smoothing procedures are described. A listing of the program is included.

INTRODUCTION

In order to furnish the necessary initial conditions in grid point form to the Subsynoptic Advection Model (SAM) developed by Glahn, Lowry, and Hollenbaugh [3], a computer program was developed to perform objective analyses with small-scale detail, based on the method of Bergthórsson and Döös [1] and Cressman [2]. This program is used operationally at the National Meteorological Center (NMC) in the analysis of sea-level pressure and saturation deficit on a grid with a grid-length of about 50 miles. An integral part of the analysis system, necessary for any operationally oriented program, is the error-detecting routine. The analysis of other scalar variables, the use of grids of different size located in other parts of the world, or the use of different first guess fields would require the redefinition of certain constants and possibly other minor changes; however, the basic program could probably be used with much less effort than would be expended in starting completely fresh.

ANALYSIS AREA AND DATA

The analysis area, shown in Fig. 1, is covered by a 39 x 40 grid which has a grid-length of about 50 miles, exactly 1/4 that used at NMC for the operational large-scale forecast models. Except for the ocean areas, this region is represented by relatively dense surface weather reports. The "Airways" observations are collected directly from the teletype line by the IBM 360-40 computer at NMC and saved on magnetic tape. These data are then decoded by a program developed within TDL [4]. Sea-level pressures are available for analysis directly from the reports. Saturation deficits, however, are not directly available and are estimated from other variables.




Figure 1. The Analysis Area and 39 x 40 grid.

Regression equations have been derived which specify total column precipitable water as a function of surface dew point, weather, and clouds [6,7]. The saturation thickness is then specified as a linear function of the precipitable water estimate and station elevation. Saturation thickness, for our purposes, is defined as that thickness between the 1000- and 500-mb. levels for which precipitation will occur for a given amount of moisture between those levels. The saturation deficit is, then, the actual thickness minus the saturation thickness. The lower limit of saturation deficit is, by definition, zero. That is, the actual thickness can decrease down to the point of precipitation but then any further decrease in thickness will bring about a decrease in moisture content and accompanying decrease in saturation thickness. Once precipitation starts, the saturation thickness decreases with the thickness. When a station is receiving precipitation, the saturation deficit there is zero.

THE ANALYSIS METHOD

The basic analysis method used at NMC is still that described by Cressman [2] in 1959, although some of the details have undergone change [8]. This method is one of successive approximations and allows the use of the geostrophic or some other wind-height relationship so that the reported winds can be used together with the reported heights (or pressures) in adjusting the required "first guess" height (or pressure) analysis. Since the geostrophic relationship between observed surface winds and sea-level pressure does not hold well, it was felt that the use of winds in the sea-level pressure analysis would not be desirable. Also, no simple relationship exists between saturation deficit and wind. Therefore, in the application reported here, winds are not used.

The analysis is performed in a series of cycles or passes over the data; in this application four passes are made. On each pass, an observation occasions a change or correction to all gridpoints within a distance R , where R is called the scan radius or radius of influence. For the first pass, a first guess is required which can be a forecast, if that is available, or can be as crude as climatology or a suitable constant. R varies with the pass number and usually decreases for later passes.

Let

$A_{i,j}$ = gridpoint values of the variable,

$A_{x,y}$ = values of the variable interpolated from gridpoints to the point x,y , and

$O_{x,y}$ = observed values of the variable at the point x,y .

The value $A_{x,y}$ is found by interpolating into the current analysis; for the first pass, the current analysis will be the first guess. Then the correction, due to a particular observation at x,y , to the gridpoints within a distance R of the observation is given by

$$C = (O_{x,y} - A_{x,y}).$$

Usually more than one observation will occasion a change at a gridpoint. These corrections can be combined in this program in any one of three ways:

$$C_1 = \frac{1}{n} \sum_{i=1}^n (O_{x,y} - A_{x,y})_i \quad d_i \leq R$$

$$C_2 = \frac{1}{n} \sum_{i=1}^n \frac{R^2 - d_i^2}{R^2 + d_i^2} (O_{x,y} - A_{x,y})_i \quad d_i < R$$

$$C_3 = \frac{\frac{1}{n} \sum_{i=1}^n \frac{R^2 - d_i^2}{R^2 + d_i^2} (O_{x,y} - A_{x,y})_i}{\frac{1}{n} \sum_{i=1}^n \frac{R^2 - d_i^2}{R^2 + d_i^2}} \quad d_i < R$$

where d is the distance from the observation to the gridpoint. The method of combination is selected by a control card on each pass. In C_2 and C_3 the corrections are weighted such that the closer the station to the gridpoint, the larger the weight. The effect of using C_3 instead of C_2 is a greater change being made at the gridpoint and more rapid convergence to the general level of the data within the influence circle.

ERROR-DETECTING PROCEDURE

This program is designed to function with no human intervention; the decision of whether or not an observation is in error is made objectively. Error detecting is of two types. The first type, used on the first pass only, employs the "buddy system". For each observation O (the subscripts x, y are omitted for convenience), the two closest neighbors with observations O_i , $i = 1, 2$, are found. Then the observation O is considered to be questionable if

$$\begin{aligned} |O - O_i| &> ER_2 & d' &\leq 1 \\ \frac{|O - O_i|}{d'} &> ER_3 & d' &> 1 \end{aligned}$$

where d' is the distance between the stations in grid units. ER_2 and ER_3 are specified by control card and are a function of the pressure in the area when pressure is being analyzed.

The observation 0 is permanently discarded if both neighbor stations indicate it to be questionable. If neither of the neighbor stations indicate 0 to be questionable, it is used on the first pass. If exactly one of the two neighbor stations indicates 0 to be questionable, the station is used on the first pass if

$$|0 - A| \leq ER_1$$

and permanently discarded otherwise, where ER_1 is specified by control card and A is the interpolated value of the field at the station location.

On each pass after the first, the observation is used on that particular pass only if

$$|0 - A| \leq ER_1$$

Specific values of ER_1 , ER_2 , and ER_3 are given below.

SMOOTHING

Usually some type of smoothing is done after one or more of the data passes. The smoothing operator employed in this program is that used by the Travelers Research Center[10] and is a generalization of the one given by Cressman [2]:

$$S_{i,j} = \frac{A_{i,j} + b\bar{A}_{i,j}}{1 + b}$$

where

$$\bar{A}_{i,j} = 1/4[A_{i+1,j} + A_{i-1,j} + A_{i,j+1} + A_{i,j-1}]$$

and $S_{i,j}$ is the smoothed gridpoint value.

Specific values of b used are given below.

SEA-LEVEL PRESSURE ANALYSIS

In the areas of dense data coverage, the first guess will influence the final analysis very little. However, in the ocean areas where hourly data are nonexistent the first guess will be largely retained. In operational use, the 6-hour 1000 mb. forecast produced by the NMC Primitive Equation (PE) model [9] is converted to sea-level pressure and used as the first guess. Values of ER_1 , ER_2 , ER_3 , R , type of correction, and b which give good results are given in Tables 1 and 2.

Table 1. Program parameters which give good results in sea-level pressure analysis.

Pass No.	ER_1 (mb)	R (grid units)	Type of Correction	b
1	14.0	8.0	C_2	0
2	11.6	5.0	C_3	5
3	5.0	2.5	C_3	1
4	3.0	1.0	C_3	2

Table 2. Values of ER_2 and ER_3 used on pass No. 1 which give good results for Sea-Level Pressure Analysis

Approximate Pressure at Station (mb)	ER_2 (mb)	ER_3 (mb/grid unit)
945-955	17	17
955-965	16	16
965-975	15	15
975-985	14	14
985-995	13	13
995-1005	11	11
1005-1015	9	9
1015-1025	7	7
1025-1035	5	5
1035-1045	3	3
1045-1055	3	3
1055-1065	2	2
1065-1075	2	2

The pressures given in Table 2 are taken from the current analysis in the vicinity of the observation. The range is more than sufficient to accommodate any observable pressure.

The use of a large scan radius is equivalent to performing a rather heavy smoothing; therefore, the analysis resulting from the first pass does not need additional smoothing. Also, the use of a relatively large scan radius on the first two passes smoothes out possible discontinuities along the coastlines where the data density decreases abruptly to zero over the oceans.

Since the first guess is usually rather good, less than 4 passes could be used if some manual rulings on the correctness of the data were possible. However, the objective decision-making procedure works extremely well with the program parameters given in Tables 1 and 2.

An example of a pressure analysis is shown in Fig. 2. Two pressure observations were discarded on the third and fourth passes as being incorrect--Muscle Shoals, Alabama (1011.0), and Timmins, Ontario (1017.9).




Figure 2. An example of sea-level pressure analysis. Data are for 0700 GMT, January 9, 1969.

SATURATION DEFICIT ANALYSIS

A field with large areas of zeros or other single value presents special problems in analysis, especially when the gradients are large near these areas. In the analysis of saturation deficit, the zero values are correct by definition and it is particularly desirable that the analyzed areas of zero contain the zero observations. It was found that certain transformations were necessary to achieve the desired results. These are shown in Fig. 3.

The first guess field is derived from the 6-hour forecast of mean relative humidity for the 1000 to 500-mb layer produced by the NMC PE model valid one hour earlier than the observation time. The relationship, derived from data furnished by Mr. Russell Younkin to Kulawiec [5] (see p. 57) is

$$FG = .0245 RH^2 - 8.05 RH + 560$$

where FG means first guess and RH indicates relative humidity.

The data given by Kulawiec [5] pertain to the mean relative humidity of an actual atmosphere. The values used within the PE model and stored on tape are "model" relative humidities with a range of 0 to 100 percent; the above equation is in terms of those model relative humidities. Setting to -15 (indicating precipitation) values of FG which are ≤ 20 , as shown in Fig. 3, is equivalent to assuming precipitation with a model relative humidity of about 94 percent; this approximates the procedure used in the PE model.

The transformed values, denoted by primes, are then used by the analysis program. The program constants are shown in Table 3.

Table 3. Program parameters which give good results in saturation deficit analysis.

Pass No.	ER ₁	ER ₂	ER ₃	R	Type of Correction	b
1	60	60	60	8	C ₂	0
2	65	--	--	4	C ₂	0
3	60	--	--	2	C ₃	1
4	55	--	--	1	C ₃	4

Estimates of saturation deficit and the transformed values are quite variable from station to station. In order not to discard estimates based on correct observations, quite liberal acceptance values have to be used. As a result, the error detecting procedure is not very effective.

During the smoothing of the saturation deficit field after the fourth pass, a gridpoint value is not altered if it is ≤ 3.5 . This special treatment helps to preserve the areas indicating precipitation.




Figure 3. The transformations used in modifying the regression estimates of saturation deficit S_d and first guess values FG before analysis.

After the analysis is completed, the gridpoint values are transformed back to their proper scale by:

$$\begin{array}{lll}
 S_d = 6S'_d - 30 & \text{if} & S'_d > 6 \\
 S_d = 0 & \text{if} & S'_d < 0 \\
 S_d = S'_d & \text{if} & 0 \leq S'_d \leq 6
 \end{array}$$

An example of saturation deficit analysis is shown in Fig. 4. Isolines are labeled in meters and have been traced from a gridprinted map at intervals of 30 meters. Actual values of S_d are not shown at the stations but stations reporting precipitation are indicated by a dot (•) and stations reporting no precipitation are shown by a square around an x (⊠). With few exceptions, the integrity of the zero S_d values is maintained.




Figure 4. An example of saturation deficit analyses. Stations reporting precipitation are indicated by dots and stations reporting no precipitation are indicated by an x inside a square. Data are for 0800 GMT, December 9, 1966.

REFERENCES

1. Bergthorssen, P., and B. R. Döös, "Numerical Weather Map Analysis," Tellus, Vol. 7, No. 3, August 1955, pp. 329-340.
2. Cressman, G. P., "An Operational Objective Analysis System," Monthly Weather Review, Vol. 87, No. 10, October 1959, pp. 367-374.
3. Glahn, H. R., D. A. Lowry, and G. W. Hollenbaugh, "An Operational Subsynoptic Advection Model," ESSA Technical Memorandum WBTM TDL 23, July 1969, 26 pp.
4. Hollenbaugh, G. W., H. R. Glahn, and D. A. Lowry, "Automatic Decoding of Hourly Weather Reports," ESSA Technical Memorandum WBTM TDL 21, July 1969, 27 pp.
5. Kulawiec, M.H., "Local Cloud and Precipitation Forecast Method (SLYH)," ESSA Weather Bureau Technical Note 13-FCST-2, September 1965, 59 pp.
6. Lowry, D. A., and H. R. Glahn, "Integrated Moisture-Surface Variable Relationships," Abstract in Bulletin of the AMS, Vol. 48, No. 3, March 1967, p. 205.
7. Lowry, D. A., and H. R. Glahn, "Relationships Between Integrated Atmospheric Moisture and Surface Weather," Journal of Applied Meteorology, Vol. 8, No. 5, October 1969, (in press).
8. McDonell, J. E., "Notes on NMC Objective Analysis Programs," and attachments, Unpublished Manuscript, March 1968.
9. Shuman, F. G., and J. B. Hovermale, "An Operational Six-Layer Primitive Equation Model," Journal of Applied Meteorology, Vol. 7, No. 4, August 1968, pp. 525-547.
10. Thomasell, A., Jr., and J. G. Welsh, "Objective Analysis of Sea-Level Pressure and Surface Temperature, Dew Point and Wind," The Travelers Research Center, Inc., Tech. Publication 19, Contract FAA/BRD-363, November 1962, 87 pp.

APPENDIX

Contained in this appendix are the listings of the three subroutines in FORTRAN language necessary for performing the analyses described. These subroutines are in the form necessary for use with the SCOPE operating system on the CDC 6600 at Suitland, Md.

Subroutine BCD is called when an analysis is desired and it in turn calls ESP, the error sensing procedure, and INTR for interpolation. Comment cards describe the program parameters.

Upon entry to BCD, certain information must be present in COMMON:

P(39,40) : First guess field
DATA(600) : Data to be analyzed
LTAG(600) : If a datum is known to be incorrect, the
corresponding value in LTAG =1; otherwise
LTAG = 0
XP(600) : x-position of corresponding datum in DATA
YP(600) : y-position of corresponding datum in DATA
NODATA : Number of datum values in DATA, maximum of 600

The lower left corner of the grid is considered to be position (1,1), with x increasing to 40 to the right and y increasing to 39 upward.

The subroutine OUP2 called by BCD is used for outputting maps. The calling statement can be eliminated or another subroutine substituted. Upon return from BCD to the calling program, the analysis is contained in P(39,40) in COMMON.

```

SUBROUTINE BCD(NCAL)
C      D413A
C      JULY, 1968      GLAHN, HOLLENBAUGH
C      BCD ANALYSIS PROGRAM
C      BERGTHORSSSEN-CRESSMAN-DOOS METHOD
C      INPUT CONTROL INFORMATION
C      NCAL=1 FOR SEA LEVEL PRESSURE ANALYSIS, GREATER THAN 1 FOR
C      SATURATION DEFICIT ANALYSIS
C      NP=NUMBER OF PASSES
C      R=RADIUS OF INFLUENCE
C      NTYPE=KIND OF CORRECTION
C          1 MEANS W=1
C          2 MEANS  $W=(R**2-D**2)/(R**2+D**2)$ 
C          3 MEANS SAME AS 2 EXCEPT SUM OF WTS IN DENOMINATOR
C      B=SMOOTHING PARAMETER (IF B=0, NO SMOOTHING DONE)
C      NCHECK=0 IF NO DATA CHECKING DONE
C      ER1=MAXIMUM ERROR FROM LAST PASS (OR FIRST GUESS)
C      ER2( )=MAXIMUM GRADIENT ALLOWABLE TO CLOSEST STATION IF
C      DISTANCE TO STATION IS GREATER THAN 1 GRID INTERVAL
C      ER3( )=MAXIMUM ALLOWABLE DIFFERENCE TO CLOSEST STATION IF
C      DISTANCE TO CLOSEST STATION IS LESS THAN 1 GRID INTERVAL
C      DSTA( )HOLDS STATION DATA NAMES
C      DATA( )HOLDS STATION SLP DATA
C      NODATA IS NUMBER OF DATA POINTS
C      XP( )HOLDS STATION DATA X-POSITIONS
C      YP( )HOLDS STATION DATA Y-POSITIONS
C      LTAG( )HOLDS INDICATORS FOR DATA ACCEPTANCE
C      P( , )HOLDS CURRENT ANALYSIS
C      CORR( , )HOLDS CORRECTIONS TO BE MADE ON CURRENT PASS
C      COUNT( , )HOLDS NUMBER OF STATIONS AFFECTING GRIDPOINT ON
C      CURRENT PASS
C      DIMENSION LTAG(600),DATA(600),DSTA(600),XP(600),YP(600),ER2(13),
C      ER3(13),P(39,40),H(1564),CORR(39,40),COUNT(39,40)
C      EQUIVALENCE (H,P)
C      COMMON/BLOCK6/CORR,COUNT,DUMMY(3351)
C      COMMON/BLOCK7/LTAG,DATA,DSTA,XP,YP,ER2,ER3,ER1,R,NODATA
C      COMMON/BLOCK8/H
C      *****
160  READ 161,NP
161  FORMAT(I4)
    DO 280 LP=1,NP
C      LP IS THE NUMBER OF THE DATA PASS
    READ 172,R,NTYPE,B,NCHECK,ER1
172  FORMAT(F4.0,I4,F4.0,I4,F4.0)
174  FORMAT(1H0I4,F5.1,I4,F5.1,I4,F5.1)
C      READ IN MAXIMUM ALLOWABLE DIFFERENCES ON FIRST PASS ONLY
    IF(LP-1)176,176,190
176  READ 177,(ER2(K),K=1,13)
    READ 177,(ER3(K),K=1,13)
177  FORMAT(13F4.0)
190  CALL ESP(LP,NCAL)
    RSQ=R*R
    DO 199 J=1,40
    DO 199 L=1,39

```

```

CORR(L,J)=0.
COUNT(L,J)=0.
199 CONTINUE
DO 240 K=1,NODATA
NSW=0
IF(LTAG(K))240,202,240
202 JB=XP(K)-R+.999
IF(JB)204,204,206
204 JB=1
GO TO 208
206 IF(JB-40)208,208,240
208 JE=XP(K)+R+.001
IF(JE)240,240,212
212 IF(JE-40)214,214,213
213 JE=40
214 LB=YP(K)-R+.999
IF(LB)216,216,218
216 LB=1
GO TO 220
218 IF(LB-39)220,220,240
220 LE=YP(K)+R+.001
IF(LE)240,240,224
224 IF(LE-39)226,226,225
225 LE=39
226 DO 237 J=JB,JE
DO 237 L=LB,LE
DISTSQ=(FLOATF(J)-XP(K))**2+(FLOATF(L)-YP(K))**2
IF(DISTSQ-RSQ)227,227,237
227 IF(NSW)228,228,231
228 NSW=1
CALL INTR(YP(K),XP(K),BB)
C SUBROUTINE INTR INTERPOLATES IN FIELD P(39,40) TO POSITION
C YP(K),XP(K) AND RETURNS ANSWER IN BB
231 IF(NTYPE-1)232,232,235
232 CORR(L,J)=CORR(L,J)+DATA(K)-BB
233 COUNT(L,J)=COUNT(L,J)+1.
GO TO 237
235 WT=(RSQ-DISTSQ)/(RSQ+DISTSQ)
CORR(L,J)=CORR(L,J)+WT*(DATA(K)-BB)
IF(NTYPE-2)233,233,2360
2360 COUNT(L,J)=COUNT(L,J)+WT
237 CONTINUE
240 CONTINUE
DO 245 L=1,39
DO 245 J=1,40
IF(COUNT(L,J))245,245,244
244 P(L,J)=P(L,J)+CORR(L,J)/COUNT(L,J)
245 CONTINUE
IF(NCAL-1)2460,2460,247
C SUBROUTINE OUP2 IS FOR OUTPUTTING MAP IN P(,)
2460 CALL OUP2(P,10.,0.,0,20,7,LP*2+20)
PRINT 174,LP,R,NTYPE,B,NCHECK,ER1
GO TO 248
247 CALL OUP2(P,1.,0.,0,30,7,LP*2+29)

```



```

PRINT 174,LP,R,NTYPE,B,NCHECK,ERI
248 IF(B)280,280,251
251 BP1=B+1.
DO 270 L=1,39
DO 270 J=1,40
GO TO(253,253,253,2510),LP
C   TAILORED TO 4 PASSES
2510 IF(P(L,J)+3.5)252,252,253
C   DURING ANALYSIS OF SEA LEVEL PRESSURE (IN MB) BRANCH TO 253
C   WILL ALWAYS OCCUR-DURING ANALYSIS OF SATURATION
C   DEFICIT BRANCH TO 252 WILL OCCUR IN AREAS OF PRECIPITATION
252 CORR(L,J)=P(L,J)
GO TO 270
253 SUM=0.
KT=0
K5=J-1
IF(K5)258,258,256
256 KT=KT+1
SUM=SUM+P(L,K5)
258 K5=J+1
IF(K5-40)260,260,262
260 SUM=SUM+P(L,K5)
KT=KT+1
262 K5=L-1
IF(K5)266,266,264
264 KT=KT+1
SUM=SUM+P(K5,J)
266 K5=L+1
IF(K5-39)268,268,269
268 KT=KT+1
SUM=SUM+P(K5,J)
269 CORR(L,J)=(P(L,J)+B*SUM/FLOAT(KT))/BP1
270 CONTINUE
DO 271 L=1,39
DO 271 J=1,40
P(L,J)=CORR(L,J)
271 CONTINUE
IF(NCAL-1)276,276,278
C   SUBROUTINE OUTP2 IS FOR OUTPUTTING MAP IN P(,)
276 CALL OUTP2(P,10.,0.,0,20,7,LP*2+21)
GO TO 280
278 CALL OUTP2(P,1.,0.,0,30,7,LP*2+30)
280 CONTINUE
290 RETURN
END

```

```

SUBROUTINE ESP(LP,NCAL)
C      D414A
C      JULY, 1968      GLAHN, HOLLENBAUGH
C      ESP=ERROR SENSING PROCEDURE
C      STATIONS NOT IN DIRECTORY ARE NOT USED.
C      IF A STATION DOES NOT AGREE WITH EITHER OF ITS TWO CLOSEST
C      NEIGHBORS IT IS PERMANENTLY DISCARDED.
C      IF A STATION DOES NOT AGREE WITH EXACTLY ONE OF ITS TWO
C      CLOSEST NEIGHBORS IT IS CHECKED WITH THE CURRENT ANALYSIS.
C      IF IT AGREES WITH ANALYSIS IT IS ACCEPTED.
C      IF IT DOES NOT AGREE WITH THE ANALYSIS IT IS PERMANENTLY
C      DISCARDED.
C      IF A STATION AGREES WITH BOTH OF ITS TWO CLOSEST NEIGHBORS
C      IT IS USED IN THE FIRST PASS.
C      ON ALL PASSES AFTER THE FIRST THE STATION IS CHECKED WITH
C      THE ANALYSIS AND USED FOR THAT PASS IF IT AGREES AND IS
C      NOT USED IN THAT PASS IF IT DISAGREES.
C      LTAG( ) USE
C      +1=PERMANENTLY DISCARDED
C      0=USE ON THIS PASS
C      -1=DO NOT USE ON THIS PASS
C      DIMENSION DSTA(600),DATA(600),H(1564),P(39,40),ER2(13),ER3(13),
1XP(600),YP(600),LTAG(600),SAVE(2,2)
C      EQUIVALENCE (H,P)
C      COMMON/BLOCK7/LTAG,DATA,DSTA,XP,YP,ER2,ER3,ER1,R,NODATA
C      COMMON/BLOCK8/H
C      *****
100  IF(LP-1)1000,1000,160
1000 DO 150 K=1,NODATA
    IF(LTAG(K))102,102,150
102  SAVE(1,1)=999999.
    SAVE(2,1)=999999.
C      FIND NEAREST TWO STATIONS TO A DATA POINT
    DO 110 L=1,NODATA
    IF(LTAG(L))1022,1022,110
1022 IF(K-L)103,110,103
103  DISTSQ=(XP(K)-XP(L))**2+(YP(K)-YP(L))**2
    IF(SAVE(2,1)-DISTSQ)110,110,105
105  IF(SAVE(1,1)-DISTSQ)106,106,108
106  SAVE(2,1)=DISTSQ
    SAVE(2,2)=DATA(L)
    GO TO 110
108  SAVE(2,1)=SAVE(1,1)
    SAVE(2,2)=SAVE(1,2)
    SAVE(1,1)=DISTSQ
    SAVE(1,2)=DATA(L)
110  CONTINUE
C      DETERMINE IF STATION IS IN SPECIFIED GRID AREA
    K1=XP(K)+.5
    IF(K1-1)117,118,118
117  K1=1
    GO TO 120
118  IF(K1-40)120,120,119
119  K1=40

```

```

120 K2=YP(K)+.5
    IF(K2-1)122,123,123
122 K2=1
    GO TO 125
123 IF(K2-39)125,125,124
124 K2=39
C     PICK UP GRID POINT VALUE (FIRST GUESS) CORRESPONDING TO
C     STATION DATA POINT
125 IF(NCAL-1)1251,1251,1250
1250 K3=1
    K4=0
    GO TO 127
1251 K3=P(K2,K1)/10.-93.5
    K4=0
    IF(K3-1)1261,127,1265
1261 PRINT 1262,P(K2,K1)
1262 FORMAT (25HOFIRST GUESS QUESTIONABLE,F7.1)
    K3=7
    GO TO 127
1265 IF(K3-13)127,127,1261
C     DETERMINE IF STATIONS DATA IS GOOD
127 DO 133 L=1,2
    IF(SAVE(L,1)-1.)132,132,129
129 IF(ABSF(DATA(K)-SAVE(L,2))/SQRTF(SAVE(L,1))-ER3(K3))133,133,130
130 K4=K4+1
    GO TO 133
132 IF(ABSF(DATA(K)-SAVE(L,2))-ER2(K3))133,133,130
133 CONTINUE
    IF(K4-1)136,142,138
136 LTAG(K)=0
    GO TO 150
138 LTAG(K)=1
    PRINT 140,DSTA(K),DATA(K),SAVE(1,2),SAVE(2,2)
140 FORMAT (1HOR3,F8.1,54H NOT ACCEPTED, INCONSISTENT WITH 2 NEAREST
1 NEIGHBORS,2F8.1)
    GO TO 150
142 CALL INTR(YP(K),XP(K),BB)
    IF(ABSF(DATA(K)-BB)-ER1)144,144,147
144 LTAG(K)=0
    PRINT 145,DSTA(K),DATA(K),SAVE(1,2),SAVE(2,2),BB
145 FORMAT (1HOR3,F8.1,55H ACCEPTED, INCONSISTENT WITH 1 OF 2 NEARES
1T NEIGHBORS,2F8.1,30H BUT AGREES WITH FIRST GUESS,F8.1)
    GO TO 150
147 LTAG(K)=1
    PRINT 149,DSTA(K),DATA(K),SAVE(1,2),SAVE(2,2),BB
149 FORMAT (1HOR3,F8.1,59H NOT ACCEPTED, INCONSISTENT WITH 1 OF 2 NE
1AREST NEIGHBORS,2F8.1,18H AND FIRST GUESS,F8.1)
150 CONTINUE
    RETURN
160 DO 168 K=1,NODATA
    IF(LTAG(K))1601,1601,168
1601 IF(1.-R-XP(K))1602,1602,165
1602 IF(R+40.-XP(K))165,1603,1603
1603 IF(1.-R-YP(K))1604,1604,165

```

```
1604 IF(R+39.-YP(K))165,161,161
161  CALL INTR(YP(K),XP(K),BB)
      IF(ABSF(DATA(K)-BB)-ER1)167,167,163
163  PRINT 164,DSTA(K),DATA(K),BB,LP
164  FORMAT (1H0R3,F8.1,43H  NOT ACCEPTED, INCONSISTENT WITH ANALYSIS,
1F8.1,14H  BEFORE PASS,13)
165  LTAG(K)=-1
      GO TO 168
167  LTAG(K)=0
168  CONTINUE
      RETURN
      END
```

```

SUBROUTINE INTR(BY,BX,BB)
C      D415A
C      JULY, 1968      GLAHN, HOLLENBAUGH
C      PERFORMS BI-QUADRATIC INTERPOLATION WHERE POSSIBLE, LINEAR
C      INTERPOLATION IN OUTSIDE GRID INTERVAL, AND LINEAR
C      EXTRAPOLATION OUTSIDE GRID
C      ARGUMENTS
C      BY=Y-COORDINATE, FROM BOTTOM
C      BX=X-COORDINATE, FROM LEFT
C      BB=INTERPOLATED (OR EXTRAPOLATED) VALUE RETURNED TO CALLING
C      PROGRAM
      DIMENSION H(1564),P(39,40),B(4)
      EQUIVALENCE (H,P)
      COMMON/BLOCK8/H
C      *****
104  NBX=BX
      NBY=BY
      IF(NBX-1)114,120,111
111  IF(NBX-39)112,120,115
112  IF(NBY-1)121,130,113
113  IF(NBY-38)140,130,123
114  NBX=1
      GO TO 120
115  NBX=39
120  IF(NBY-1)121,130,122
121  NBY=1
      GO TO 130
122  IF(NBY-39)130,123,123
123  NBY=38
C      STATEMENT 130 STARTS BI-LINEAR INTERPOLATION-EXTRAPOLATION
130  NBXP1=NBX+1
      NBYP1=NBY+1
      DX=BX-FLOATF(NBX)
      DY=BY-FLOATF(NBY)
      BB=P(NBY,NBX)+(P(NBY,NBXP1)-P(NBY,NBX))*DX+(P(NBYP1,NBX)-
1P(NBY,NBX))*DY+(P(NBY,NBX)+P(NBYP1,NBXP1)-P(NBYP1,NBX)-P(NBY,
2NBXP1))*DX*DY
      RETURN
C      STATEMENT 140 STARTS BI-QUADRATIC INTERPOLATION
140  DX=BX-FLOATF(NBX)
      DY=BY-FLOATF(NBY)
      NBYP2=NBY+2
      NBYP1=NBY+1
      NBYM1=NBY-1
      FCT=(DY**2-DY)/4.
      FET=(DX**2-DX)/4.
      DO 145 J=1,4
      N=NBX-2+J
      B(J)=P(NBY,N)+(P(NBYP1,N)-P(NBY,N))*DY+(P(NBYM1,N)+P(NBYP2,N)-
1P(NBY,N)-P(NBYP1,N))*FCT
145  CONTINUE
      BB=B(2)+(B(3)-B(2))*DX+(B(1)+B(4)-B(2)-B(3))*FET
      RETURN
      END

```

(Continued from inside front cover)

- WBTM TDL 16 Objective Visibility Forecasting Techniques Based on Surface and Tower Observations. Donald M. Gales, October 1968. (PB-180 479)
- WBTM TDL 17 Second Interim Report on Sea and Swell Forecasting. N. A. Pore and Lt. W. S. Richardson, USESSA, January 1969. (PB-182 273)
- WBTM TDL 18 Conditional Probabilities of Precipitation Amounts in the Conterminous United States. Donald L. Jorgensen, William H. Klein, and Charles F. Roberts, March 1969. (PB-183 144)
- WBTM TDL 19 An Operationally Oriented Small-Scale 500-Millibar Height Analysis. Harry R. Glahn, March 1969.
- WBTM TDL 20 A Comparison of Two Methods of Reducing Truncation Error. Robert J. Bernowitz, May 1969. (PB-184 741)
- WBTM TDL 21 Automatic Decoding of Hourly Weather Reports. George W. Hollenbaugh, Harry R. Glahn, and Dale A. Lowry, July 1969.